
Stichwort-Erkennung

�bersicht

Sprachverarbeitung ohne Syntax?

System erkennt einzelne Stichw�rter, ignoriert den Rest

Quelle: [Covington, 1994]

Beantworten von nat�rlichsprachlichen (englischen) Anfragen zu einer Datenbank

Datenbank-Konzepte

Aufbau des Programms

Lambda-Abstraktion und Beta-Reduktion in Prolog

einzelne Pr�dikate

Zweck

Vorstellen des Programms; kein Pr�fungsstoff

Stichwort-Erkennung

Viele Systeme versuchen gar nicht erst, die Struktur der 

Eingabes�tze herauszufinden.

Stattdessen erkennen sie einige wenige Stichw�rter 

(keyword spotting).

sehr effizient

kein grosser Entwicklungsaufwand

robust: Auch unerwartete/fehlerhafte/É Eingaben bringen das Sys-

tem nicht durcheinander

aus linguistischer Sicht v�llig unangemessen

grosse Gefahr, dass Eingaben falsch verarbeitet werden

Beispiel: Datenbank-Zugriff

Als Beispiel f�r ein Keyword-Spotting-System bauen wir 

eine nat�rlich-sprachliche Schnittstelle zu einer Daten-

bank.

Which women work

for our company?
Who is the boss?

Please tell me whether there is any

male programmer whose age is over 40.

Fire the under-

paid men!

Is anyoneÕs salary

under 25000?



Stichwort-Erkennung Ð 4

Relationale Datenbanken

Relationale Datenbanken bestehen aus Tabellen:

ID Name Age Sex Title Salary

1001 Doe, John 47 m President 100 000

1002 Smith, Mary 28 f Programmer 60 000

1003 Zimmer, Fred 35 m Sales Representative 55 000

1004 Smith, Mark 48 m Programmer 15 000

1005 Russell, Ann 31 f Programmer 17 000

Schl�ssel-Attribut: Eindeutige Identifikation eines Eintrags

Relationale Datenbanken

Typischerweise umfasst eine relationale Datenbank

mehrere Tabellen.

ID Name Age Sex Title Salary

1001 Doe, John 47 m President 100 000

1002 Smith, Mary 28 f Programmer 60 000

1003 Zimmer, Fred 35 m Sales Representative 55 000

1004 Smith, Mark 48 m Programmer 15 000

1005 Russell, Ann 31 f Programmer 17 000

Chef Untergeben

1001 1002

1001 1003

1002 1004

1002 1005

Wir beschr�nken uns hier der Einfachheit halber jedoch 

auf eine einzige Tabelle.

Relationale Datenbanken

Die meisten relationalen Datenbanken werden �ber die 

Abfragesprache SQL angesteuert.

SELECT NAME, AGE
  FROM EMPLOYEE
  WHERE SEX = 'm'
    AND TITLE = 'Programmer'
    AND AGE > 40

Statt mit SQL zu arbeiten, basteln wir uns schnell eine 

eigene (sehr stark vereinfachte) Datenbankverwaltung in 

Prolog.

ÈDatenbankÇ in Prolog

Der Einfachheit halber repr�sentieren wir die einzige 

Tabelle als Prolog-Fakten:

employee(1001, 'Doe, John',    47, m, 'President', 100000).
employee(1002, 'Smith, Mary',  28, f, 'Programmer', 60000).
employee(1003, 'Zimmer, Fred', 35, m, 'Sales Rep',  55000).
employee(1004, 'Smith, Mark',  48, m, 'Programmer', 15000).
employee(1005, 'Russell, Ann', 31, f, 'Programmer', 17000).

ID Name Age Sex Title Salary

1001 Doe, John 47 m President 100 000

1002 Smith, Mary 28 f Programmer 60 000

1003 Zimmer, Fred 35 m Sales Representative 55 000

1004 Smith, Mark 48 m Programmer 15 000

1005 Russell, Ann 31 f Programmer 17 000

ÈDatenbankÇ in Prolog

Die Abfrage geschieht �ber Prolog-Regeln:

employee(1001, 'Doe, John',    47, m, 'President', 100000).
employee(1002, 'Smith, Mary',  28, f, 'Programmer', 60000).
employee(1003, 'Zimmer, Fred', 35, m, 'Sales Rep',  55000).
employee(1004, 'Smith, Mark',  48, m, 'Programmer', 15000).
employee(1005, 'Russell, Ann', 31, f, 'Programmer', 17000).

salary(ID, Salary) :-
  employee(ID, _, _, _, _, Salary).

age(ID, Age) :-
  employee(ID, _, Age, _, _, _).



Stichwort-Erkennung Ð 9

ÈDatenbankÇ in Prolog

Eintr�ge werden mittels retract gel�scht:

:- dynamic(employee/6).

employee(1001, 'Doe, John',    47, m, 'President', 100000).
employee(1002, 'Smith, Mary',  28, f, 'Programmer', 60000).
employee(1003, 'Zimmer, Fred', 35, m, 'Sales Rep',  55000).
employee(1004, 'Smith, Mark',  48, m, 'Programmer', 15000).
employee(1005, 'Russell, Ann', 31, f, 'Programmer', 17000).

retract(employee(1003,_,_,_,_,_)).

Das 6stellige employee kann w�hrend 

des Programmablaufs ver�ndert werden.

In unserem Beispiel werden nie Eintr�ge ge�ndert

oder neu eingef�gt.

Arbeitsweise

Die W�rter lassen sich in verschiedene Kategorien

unterteilen:

Aktionen Ñ show, delete, fire, tell

Eigenschaften/Tests Ñ male, woman, underpaid, programmer

Vergleiche Ñ over, under

Argumente f�r Vergleiche Ñ salary, age, 30000

Unbekannte W�rter Ñ please, me, whether, elephant

Please tell me whether there is any male

programmer whose age is over 40.



Stichwort-Erkennung Ð 11

Arbeitsweise

Please tell me whether there is any
male programmer whose age is over 40.

[action(display_records),
 test(A^sex(A,m)),
 test(B^title(B,Programmer)),
 argument(C^D^age(C,D)),
 comparison(E^F^(E>F)),
 argument(_^G^(G=40))]

[please,tell,me,whether,there,
 is,any,male,programmer,whose,
 age,is,over,40,'.']

sex(X,m),
title(X,'Programmer'),
age(X,Y), Z = 40, Y > Z,
employed(X)

[display_records]

[1004,1009]

display_records(
  [1004,1009])

findall

Tokenizer

filter

translate

=..

Arbeitsweise

Das Programm:

bittet den Benutzer, eine Anfrage zu stellen

liest die Benutzereingabe ein und zerlegt sie in einzelne Tokens

filtert alle unbekannten W�rter heraus

Þ ÈBedeutungÇ der bekannten Stichw�rter

geht die Liste der ÈBedeutungenÇ durch und bestimmt

welchen Bedingungen ein X gen�gen muss, um zur L�sungsmenge zu geh�ren

welche Aktionen auf diese X angewandt werden

f�hrt die Aktionen f�r die L�sungsmenge aus

testet, ob die Eingabe ÈquitÇ war

nein: Backtracking bis ganz an den Anfang

ja: Ende

Arbeitsweise

Unbekannte W�rter werden einfach ignoriert.

Das ist nicht unproblematisch:

ÒPaul is happyÒ ­ ÒPaul is not happyÓ

Bekannten W�rtern wird eine Bedeutung zugewiesen.

Was ist die Bedeutung eines Wortes?

Was ist mit Mehrdeutigkeiten?

Die Struktur des Satzes wird v�llig ignoriert.

Programm-Rumpf

:- ensure_loaded(tokenizer).

process_queries :-
repeat,

write('Query: '),
read_atomics(Words),
process_query(Words),

Words == [quit],
!.

process_query(Query) :-
filter(Query, Filtered),
translate(Filtered, X, TestsOnX, Actions),
findall(X, TestsOnX, MatchingEmployees),
execute_actions(Actions, MatchingEmployees).

L�dt das Programm in tokenizer.pl; die 

dort definierten Pr�dikate sind danach 

benutzbar.

Konstruktion der Abfrage

Welche Tupel sollen nun angezeigt werden?

die m�nnlichen Programmierer, deren Alter �ber 40 liegt

jene X

deren Geschlecht m�nnlich ist

deren Titel Programmierer ist

und deren Alter Y gr�sser als 40 ist

sex(X,m), title(X,programmer),
age(X,Y), Z = 40, Y > Z

Please tell me whether there is any male

programmer whose age is over 40.

Lambda-Abstraktion

Im Lambda-Kalk�l k�nnen Formeln in Eigenschaften 

umgewandelt werden.

sterblich ( fido )

Fido ist sterblich

AUSSAGE

l x sterblich ( x )

die Eigenschaft, sterblich zu sein

EIGENSCHAFT

sterblich ( x )

X ist sterblich

Abstraktion von

einem konkreten
Objekt



Stichwort-Erkennung Ð 17

Lambda-Abstraktion

Weil ein Lambda-Ausdruck als Funktion interpretiert 

wird, kann er ein Argument erhalten und liefert ein 

Ergebnis:

[[ l x  sterblich ( x ) ]] ={ áfido, wahrñ,
álimmat, falschñ,
ásokrates, wahrñ}

l x sterblich ( x )

die Eigenschaft, sterblich zu sein

EIGENSCHAFT

( fido

ATOM

)
[[ ( l x  sterblich ( x ) ) ( fido ) ]] = wahr

Beta-Reduktion

Das Argument kann auch rein syntaktisch eingesetzt wer-

den (sog. Beta-Reduktion):

Jedes Vorkommen der abstrahierten Variable (hier: x) wird durch 

das Argument (hier: fido) ersetzt.

Beide Ausdr�cke werden gleich interpretiert.

Denn die Interpretation von Lambda-Ausdr�cken ist gerade so definiert, dass dieser 

Konversionsschritt m�glich wird.

Die abstrahierte Variable (hier: x) darf jedoch nicht im Argument (hier: fido) enthal-

ten sein.

sterblich ( fido )l x sterblich ( x ) (fido)

Lambda-Abstraktion in Prolog

M�gliche Darstellungen in Prolog:

Prolog kennt (anders als z.B. Lisp) keine Lambda-Ausdr�cke als 

Sprachelemente Þ jede andere Darstellung ginge genausogut

Die Funktoren k�nnten auch einen anderen Namen tragen

l x sterblich ( x )

darzustellender l-Ausdruck

lambda(X, sterblich(X))

Darstellung mit Funktor ÈlambdaÇ

X^sterblich(X)

Darstellung mit Infix-Operator È^Ç

Beta-Reduktion in Prolog

Die Beta-Reduktion kann in Prolog durch das Unifizieren 

von Termen simuliert werden:
Die Simulation stimmt nicht ganz immer; dies w�rde jedoch hier zu weit f�hren.

X^sterblich(X) fido^Ergebnis

fido^sterblich(fido)

X / fido

Ergebnis / sterblich(fido)

ÈBedeutungÇ der Stichw�rter

Wir notieren die ÈBedeutungÇ der Stichw�rter als

Lambda-Ausdr�cke:

Aktionen

keyword(show, action(display_records))

Tests
keyword(man, test(X^sex(X, m)))

Vergleiche
keyword(under, comparison(X^Y^(X < Y)))

Argumente f�r Vergleiche
keyword(salary, argument(X^Y^salary(X, Y)))

Arbeitsweise: process_query/0

process_query/0

Ruft Tokenizer auf

Ruft Hilfspr�dikat process_query/1 auf

(f�r Fehlersuche: process_query_chattering/1)

Herausfiltern unbekannter W�rter Ñ filter/2

�bersetzen der ÈBedeutungenÇ in auf�hrbare Prolog-Terme Ñ translate/4

® Liste mit Tests, welche X erf�llen muss, um zur L�sungsmenge zu geh�ren

® Liste der Aktionen, die auf die L�sungsmenge angewandt werden sollen

Bestimmen der L�sungsmenge Ñ findall/3

Anwenden der Aktionen auf L�sungsmenge Ñ execute_query/2

Nochmals von vorne, ausser die Eingabe war ÈquitÇ

Arbeitsweise: filter/2

Das Pr�dikat filter/2 besteht aus Klauseln f�r:

Abbruchbedingung

Bekanntes Stichwort

ÈBedeutungÇ nachschlagen (mit keyword/2) und zur�ckliefern

rekursiver Aufruf

Synonym eines bekannten Stichworts

Das synonyme Stichwort herausfinden (mit synonym/2)

Dessen ÈBedeutungÇ nachschlagen (mit keyword/2) und zur�ckliefern

rekursiver Aufruf

Unbekanntes Wort

rekursiver Aufruf (d.h. das unbekannte Wort wird einfach ignoriert)

Arbeitsweise: translate/4

Das Pr�dikat translate/4 besteht aus Klauseln f�r:

Abbruchbedingung

Aktion

zur Liste der auszuf�hrenden Aktionen hinzuf�gen, z.B. display_records

rekursiver Aufruf

test(l v Test)

beta-reduzierten Test zur Liste der Bedingungen hinzuf�gen, z.B. underpaid(X)

rekursiver Aufruf

argument, comparison, argument

mehrere Beta-Reduktions-Schritte

zur Liste der Bedinungen hinzuf�gen, z.B. salary(X, Y), Z=30000, Y < Z

rekursiver Aufruf

Arbeitsweise: execute_query/2

Das Pr�dikat execute_query/2:

f�r eine leere Liste von Aktionen: display_records
Wenn keine Aktion vom System erkannt wurde, ist Anzeigen wohl am ehesten gem-

eint

Aussderm ist dies sicherer als Èim Zweifelsfall entlassenÇ

f�r eine nicht-leere Liste von Aktionen:

F�r jede Aktion: Aufruf eines passenden Pr�dikats mit der L�sungsmenge

als einzigem Argument

Dadurch werden Eingaben mit mehreren Aktionen korrekt verarbeitet, z.B.

Please display all my programmers, and fire them afterwards.



keyword.pl
Programmiertechniken der Computerlinguistik 2 á Sommersemester 1999

Online unter http://www.coli.uni-sb.de/~brawer/prolog/keyword/

% ============================================================
% keyword.pl
%
% A keyword-based natural-language interface to a database
% with employees in a company.
%
% Usage:
%  1. ?- [keyword].              % consult this file
%  2. ?- process_queries.        % start
%  3. who works as a programmer? % enter your queries
%     fire all underpaid men     % etc.
%  4. quit                       % stop by entering "quit"
%
% Taken from [Covington, 1994], but with large modifications.
% ============================================================

% Ensure that the tokenizer is loaded.
:- ensure_loaded('tokenizer.pl').

% ------------------------------------------------------------
% process_queries
% ------------------------------------------------------------
% Asks the user to enter a query and processes the entered
% query.

process_queries :-
repeat,

write('Query: '),
read_atomics(Words),
process_query(Words),

Words == [quit],
!.

% ------------------------------------------------------------
% process_query(+Query)
% ------------------------------------------------------------
% Processes a single query. Call process_query_chattering in
% order to generate output which can be useful for debugging.
%
% Query: A list of tokens, e.g. [tell,me,who,is,the,boss,'.']

process_query(Query) :-
filter(Query, Filtered),
translate(Filtered, X, TestsOnX, Actions),
findall(X, TestsOnX, MatchingEmployees),
execute_query(Actions, MatchingEmployees).

process_query_chattering(Query) :-
filter(Query, Filtered),
write('Filtered: '), write(Filtered), nl,
translate(Filtered, X, TestsOnX, Actions),
write('Tests on '), write(X), write(': '), write(TestsOnX), nl,
write('Actions: '), write(Actions), nl,
findall(X, TestsOnX, MatchingEmployees),
write('Matching Employees: '), write(MatchingEmployees),
execute_query(Actions, MatchingEmployees).

% ------------------------------------------------------------
% execute_query(+Actions, +Employees)
% ------------------------------------------------------------
% If one or more action keywords were found, execute these
% actions by calling execute_actions/2. Otherwise, display
% the matching employees.
%
% Actions:   A list of actions, e.g. [display_records]
% Employees: A list with the IDs of matching employees,
%            for instance [1001, 1004, 1009]

execute_query([], Employees) :-
display_records(Employees), !.

execute_query(Actions, Employees) :-
execute_actions(Actions, Employees).

% ------------------------------------------------------------
% execute_actions(+Actions, +Employees)
% ------------------------------------------------------------
% For each element of the Actions list, a term is constructed
% which has that element as its functor and the entire Employee
% list as its single argument.  Then, call is used to invoke
% that predicate.
%
% Example: execute_actions([display_records, remove_records],
%                          [1001,1004,1009])
% call -----> display_records([1001,1004,1009])
% call -----> remove_records([1001,1004,1009])

execute_actions([], _).
execute_actions([Action|Actions], Employees) :-

Command =.. [Action, Employees],
call(Command),
execute_actions(Actions, Employees).



% ------------------------------------------------------------
% translate(+KeywordSemantics, ?Variable, -Tests, -Actions)
% ------------------------------------------------------------
% Translates a list of keyword semantics into a number of
% tests over Variable and into a number of actions to be
% performed on the employees which pass the tests.
%
% KeywordSemantics: List of keyword semantics
% Tests:            Single Prolog term which can be passed
%                   to findall in order to determine the
%                   matching employees.
% Actions:          Actions to be performed on the set of
%                   matching employees.

translate([], X, employed(X), []).

translate([action(Action) | Items], X, Tests, [Action|Actions]) :-
translate(Items, X, Tests, Actions).

translate([test(X^Test) | Items], X, (Test,Tests), Actions) :-
translate(Items, X, Tests, Actions).

translate([argument(X^Y^T1),
comparison(Y^Z^T3),
argument(X^Z^T2) | Items],
X, (T1, T2, T3,Tests), Actions) :-

translate(Items, X, Tests, Actions).

% ------------------------------------------------------------
% filter(+Token, -KeywordSemanticsList)
% ------------------------------------------------------------
% Takes a list of tokens and extracts those known as keyword
% (or as a synonym of a keyword).  The result is a list of
% the "semantics" of the recognized keywords.

% Termination
filter([], []).

% Known keyword
filter([Word|Words], [Translation|Translations]) :-

keyword(Word, Translation),
!,
filter(Words, Translations).

% Synonym
filter([Word|Words], [Translation|Translations]) :-

synonym(Word, Synonym),
keyword(Synonym, Translation),
!,
filter(Words, Translations).

% Unknown word
filter([_|Words], Translations) :-

filter(Words, Translations).

% ------------------------------------------------------------
% keyword(+Keyword, -Semantics)
% ------------------------------------------------------------
% Associates keywords with their semantics.

keyword(show, action(display_records)).
keyword(delete, action(remove_records)).
keyword(man, test(X^sex(X, m))).
keyword(woman, test(X^sex(X, f))).
keyword(president, test(X^title(X, 'President'))).
keyword(programmer, test(X^title(X, 'Programmer'))).
keyword(salesrep, test(X^title(X,'Sales Representative'))).
keyword(employee, test(X^title(X, _))).
keyword(underpaid, test(X^underpaid(X))).
keyword(friendly, test(X^friendly(X))).
keyword(unfriendly, test(X^unfriendly(X))).
keyword(over, comparison(Y^Z^(Y>Z))).
keyword(under, comparison(Y^Z^(Y<Z))).
keyword(salary, argument(X^Y^salary(X, Y))).
keyword(age, argument(X^Y^age(X, Y))).
keyword(N, argument(_^Y^(Y=N))) :- number(N).

% ------------------------------------------------------------
% synonym(+Synonym, -Keyword)
% ------------------------------------------------------------
% Associates synonyms with synonymous keywords.

synonym(boss, president).
synonym(programmers, programmer).
synonym(men, man).
synonym(male,man).
synonym(guy, man).
synonym(women, woman).
synonym(employees, employee).
synonym(female,woman).
synonym(display, show).
synonym(present, show).
synonym(give, show).
synonym(tell, show).
synonym(remove, delete).
synonym(fire, delete).



% ------------------------------------------------------------
% display_records(+EmployeeIDList)
% ------------------------------------------------------------
% Displays employees to the user.
%
% EmployeeIDList: A list of employee IDs, e.g. [1001,1004]

display_records([]).
display_records([ID|IDs]) :-

write(ID), write(': '),
full_name(ID, Name), write(Name),
write(', '), title(ID, Title), write(Title),
write(', salary '), salary(ID, Salary), write(Salary),
write('.'), nl,
display_records(IDs).

% ------------------------------------------------------------
% remove_records(+EmployeeIDList)
% ------------------------------------------------------------
% Removes employees from the database.
%
% EmployeeIDList: A list of employee IDs, e.g. [1001,1004]

remove_records([]).
remove_records([ID|IDs]) :-

retract(employee(ID, _, _, _, _, _, _)),
remove_records(IDs).

% ------------------------------------------------------------
% Database Predicates
% ------------------------------------------------------------
% Represents the database as Prolog clauses.  A real system
% would access an external database management system, e.g.
% by passing SQL queries to a DB server.

:- dynamic(employee/7).

employee(1001, 'Doe, John',    47, m, friendly,   'President', 100000).
employee(1002, 'Smith, Mary',  28, f, friendly,   'Programmer', 60000).
employee(1003, 'Zimmer, Fred', 35, m, unfriendly, 'Sales Representative',

55000).
employee(1004, 'Smith, Mark',  48, m, friendly,   'Programmer', 15000).
employee(1005, 'Russel, Ann',  31, f, unfriendly, 'Programmer', 17000).

full_name(ID, Name) :- employee(ID, Name, _, _, _, _, _).
age(ID, Age) :- employee(ID, _, Age, _, _, _, _).
sex(ID, Sex) :- employee(ID, _, _, Sex, _, _, _).
friendly(ID) :- employee(ID, _, _, _, friendly, _, _).
unfriendly(ID) :- employee(ID, _, _, _, unfriendly, _, _).
title(ID, Title) :- employee(ID, _, _,_,_, Title, _).
salary(ID, Salary) :- employee(ID, _, _,_,_, _, Salary).
underpaid(ID) :- salary(ID, Salary), Salary < 40000.
employed(ID) :- employee(ID, _,_,_,_,_,_).

